

Universidade Federal de Santa Catarina
Pós-Graduação em Inglês
Andréia S. Rauber
asrauber@gmail.com

Análise automática de formantes e plotagem de vogais no Praat

O presente tutorial objetiva detalhar os procedimentos para análise automática de formantes de vogais utilizando-se scripts do Praat (Boersma & Weenink, 2006) para tal. Após a explicação sobre a medição de vogais também serão apresentados scripts para plotar valores de F1 e F2 de vogais tanto em escala linear como em escala logarítmica (\log_{10}).

Os scripts para medição de formantes foram escritos por Paul Boersma (Universidade de Amsterdam), já os scripts para plotar vogais foram escritos por Ricardo Bion (Universidade Federal de Santa Catarina).

1. Medição automática de formantes

Para que se possa medir formantes de forma automática, o primeiro passo é segmentar as vogais alvo. Tal segmentação é um tanto confusa quando não se tem muita prática com formas de onda e espectrogramas. Levando-se tal limitação em consideração, a subseção 1.1 exemplifica como tal segmentação pode ser realizada.

1.1 Segmentação de vogais

O primeiro passo para segmentar vogais no Praat é criar um TextGrid. Para tal:

1. Abra o arquivo de som onde as vogais estão gravadas na janela Praat objects. Para abrir um arquivo de som, selecione no menu Read a opção Read from file. Localize a pasta onde tal arquivo foi salvo e clique em Open.

2. Em seguida, ainda em Praat objects, selecione no menu à direita da tela o botão Annotate e então a opção To TextGrid (ver Figura 1). Uma tela se abrirá e nela constará a opção All tier names: Mary John bell. Estes três nomes (Mary, John e bell) são nomes de tiers (ou camadas) inventados pelos autores do Praat. Cada nome, desde que separado por um espaço, representa uma tier. Você pode mudar e/ou apagar os nomes. Para este tutorial, utilizarei apenas duas tiers, por isso mantereí apenas os nomes Mary e John.




Figura 1. Botão Annotate, opção To TextGrid na janela Praat objects.

3. Uma vez criada a TextGrid, selecione os dois arquivos abertos em Praat objects. Para tal, clique em um dos arquivos e arraste o mouse para selecionar o outro arquivo, ou pressione a tecla CTRL e clique nos dois arquivos. Após selecionados os arquivos, clique no botão Edit. Uma nova janela se abrirá, onde constarão as formas de onda e o espectrograma do arquivo de som na parte superior da janela, e duas faixas (tiers) na parte inferior da janela. Caso não seja possível visualizar as formas de onda e o espectrograma, selecione uma porção menor do arquivo de som e clique no botão “sel” no canto inferior esquerdo da tela. A forma de onda e o espectrograma não são exibidos para porções maiores do que 10 segundos, por isso a necessidade de se selecionar apenas uma pequena parte do arquivo de som.

4. Criada a TextGrid e já visualizando-se a forma de onda e o espectrograma, pode-se iniciar a segmentação. Sugiro que se segmente a palavra na primeira tier e a vogal na segunda tier, mas isso depende do interesse de cada pesquisador. Para segmentar uma palavra, localize o seu início e clique uma vez na forma de onda ou no espectrograma. Em cada tier será possível observar um pequeno círculo (ver Figura 2). Clique sobre este círculo e uma marca de segmentação será inserida (duas barras verticais vermelhas). Repita o procedimento no local onde a palavra termina. Clique sobre a parte segmentada da tier e escreva o texto que quiser. Símbolos fonéticos podem ser usados de acordo com os códigos descritos no manual do Praat (para ter acesso aos códigos, selecione no menu Help a opção Phonetic symbols). Caso você tenha inserido uma marca de segmentação (boundary) em local impróprio, você pode movê-la clicando sobre a mesma e arrastando ao ponto desejado. Também é possível apagar a boundary pressionando simultaneamente as teclas ALT+Backspace.




Figura 2. Círculo que permite inserir marcas de segmentação em uma tier.

5. O mesmo procedimento será repetido para segmentar as vogais. Note que para localizar o início e o fim de uma vogal, é importante observar tanto a forma de onda como o espectrograma (ver Figura 3). Também ouvir o som segmentado auxilia muito a decidir se a segmentação foi feita adequadamente.




Figura 3. Segmentação de palavras na tier Mary e de vogais na tier John.

Observação: é importante simplificar e padronizar as labels (etiquetas) utilizadas na segmentação para facilitar a redação do script que criará automaticamente a tabela com as vogais. No arquivo-exemplo deste tutorial, as vogais estudadas são 11 vogais do inglês. Os códigos escolhidos foram (vogal – código):

/i/ - 1	/ɑ/ - 6
/ɪ/ - 1	/ɔ/ - 7
/eɪ/ - 2	/oʊ/ - 8
/ɛ/ - 3	/ʊ/ - 9
/æ/ - 4	/u/ - u
/ʌ/ - 5	

Note que cada vogal é pronunciada três vezes e por isso tanto as palavras quanto as vogais possuem números. Exemplo: seat1, seat2, seat3.

6. O próximo passo é salvar a TextGrid. Assim como nomear as labels de forma padronizada é importante, o nome do arquivo de som e do arquivo da TextGrid também são importantes para facilitar a redação do script. Os nomes dados aos arquivos deste tutorial foram:

US_M_1.wav
US_M_1.TextGrid

Como pode haver necessidade de comparar-se vogais de diferentes dialetos ou idiomas, a primeira parte do nome dos arquivos (US) representa o grupo de falantes (neste caso, falantes nativos do inglês americano). A letra “M” significa que o participante é do sexo masculino, e “1” indica o número do participante.

Para salvar a TextGrid, na janela que mostra o arquivo de som e as TextGrids, selecione no Menu File a opção Write TextGrid to text file... Selecione a pasta onde está o arquivo de som e clique em Salvar.

Após a segmentação, cria-se uma tabela geral para organizar os dados a serem analisados. O procedimento para criar esta tabela automaticamente será descrito na próxima seção.

1.2 Criar tabela de dados

A presente seção descreverá o procedimento para criar uma tabela que conterà as seguintes colunas:

Speaker – número do participante (nome do arquivo, neste caso US_M_1);
Dialect – falantes nativos do inglês Americano (US);
Gender – neste caso, o falante é do sexo masculino (M);
Vowel – código para cada vogal, de acordo com o Praat;
Phrase – número da repetição (de um total de 3);
Cons – contexto fonológico onde a vogal está inserida;
Start – ponto de início da segmentação;
End – ponto final da segmentação;
Dur – duração do trecho segmentado;
F0 – frequência fundamental;
F1, F2, F3 – primeiro, segundo e terceiro formantes;
B1, B2, B3 – largura de banda de cada formante.

Para evitar erros de digitação do caminho onde o arquivo de som e a TextGrid estão salvos, uma sugestão é copiar os scripts a serem usados para a pasta onde foram salvos o arquivo de som e

a TextGrid. Assim, não é necessário digitar o nome das pastas e sub-pastas, já que o script buscará as informações no diretório onde está salvo.

Para abrir o script abaixo, na janela Praat objects, clique no menu Praat, opção Open Praat script... Selecione o script "CreateTable5600.praat" e clique em Open.

No script, alguns dados devem ser alterados para que o mesmo rode com arquivos de som e TextGrids diferentes dos deste tutorial. Os dados a serem alterados estão assinalados em negrito e um comentário foi inserido numa caixa de texto.

Script 1 – criar tabela

```
# Praat script CreateTable5600.praat
# Paul Boersma, April 25, 2006

Create Table with column names... table5600 162
... speaker dialect gender vowel phrase cons
... start end dur F0 F1 B1 F2 B2 F3 B3

row = 0
```

162 – número de linhas da tabela. Este número é encontrado dividindo-se o total de intervalos da TextGrid por 2. Por que dividir por dois? O Praat considera cada intervalo como uma label, mas nem todos os intervalos contém informações que interessam para a análise. O script buscará apenas as labels que tenham algo escrito. Por isso, ao descartar-se as labels em branco, dividimos o total de intervalo por 2. Como saber o número de intervalos? Na janela Praat objects, selecione a TextGrid e clique no botão Query, opção Get number of intervals. Especifique a tier (2, neste caso). O número de intervalos é de 325. Este número dividido por 2 = 162,5. Arredondando-se para baixo = 162.

```
call measureSpeakers US M 1
```

No exemplo deste tutorial, temos apenas um arquivo de som a ser analisado. Por isso o número de participantes é 1. Mas este número deve ser alterado de acordo com o número de arquivos (participantes) a serem analisados. "US" e "M" também devem ser alterados de acordo com os nomes dados aos arquivos (referente a dialeto e sexo).

Alterar o número total de linhas aqui também.

```
assert row = 162 ; 'row'
select Table table5600
Write to table file... table5600.txt
```

Definir número de intervalos = 325.

```
procedure measureSpeakers dialect$ gender$ numberOfSpeakers
  for speaker to numberOfSpeakers
    speaker$ = "'dialect$'_'gender$'_'speaker'"
    Read from file... 'speaker$'.TextGrid
    numberOfIntervals = Get number of intervals... 2
    assert numberOfIntervals = 325 ; 'speaker$'
    for iinterval to numberOfIntervals
      label$ = Get label of interval... 2 iinterval
      if label$ <> ""
        start = Get starting point... 2 iinterval
        end = Get end point... 2 iinterval
        duration = end - start
        assert duration > 0.010 ; 'speaker$' 'start'
        #
        # Get the vowel: one of i, 1, 2, 3, 4, 5, 6, 7, 8, 9, u.
        #
        vowel$ = mid$ (label$, 2, 1)

        #if vowel$ = "\"
        #  vowel$ = mid$ (label$, 4, 3)
        #endif

        if vowel$ = "1"
          vowel$ = "\ic" ; tiny fix for plotting later
        endif
        if vowel$ = "2"
          vowel$ = "e" ; tiny fix for plotting later
        endif
        if vowel$ = "3"
          vowel$ = "\ef" ; tiny fix for plotting later
        endif
        if vowel$ = "4"
          vowel$ = "\ae" ; tiny fix for plotting later
        endif
        if vowel$ = "5"
          vowel$ = "\vt" ; tiny fix for plotting later
        endif
        if vowel$ = "6"
          vowel$ = "\as" ; tiny fix for plotting later
        endif
        if vowel$ = "7"
          vowel$ = "\ct" ; tiny fix for plotting later
        endif
      endif
    endfor
  endfor
endprocedure
```

O número 2 que aparece em negrito 4 vezes refere-se ao número da tier. Como as vogais do nosso exemplo estão na tier 2, não se deve alterar este número.

Textos explicativos são precedidos de #. Ou seja, este sinal indica que a linha não será lida quando o script for rodado.

Nesta parte do script, os códigos adotados para representar cada vogal são mudados para o símbolo fonético que aparecerá na tabela e futuramente nos plots.

```

endif
if vowel$ = "8"
    vowel$ = "o" ; tiny fix for plotting later
endif
if vowel$ = "9"
    vowel$ = "\hs" ; tiny fix for plotting later
endif

#
# Get the phrasal context: one of 1 (first position), 2 (second position).
#
phrase$ = right$ (label$, 1)
assert phrase$ = "1" or phrase$ = "2" or phrase$ = "3" ; 'speaker$' 'start'
#
# Get the consonantal context (first C): one of p, t, k, s, l.
#
consonant$ = mid$ (label$, 1, 1)
assert index ("ptksl", consonant$) ; 'speaker$' 'start'
#
#Store results in table5600.
#
select Table table5600
row += 1
Set string value... row speaker 'speaker$'
Set string value... row dialect 'dialect$'
Set string value... row gender 'gender$'
Set string value... row vowel 'vowel$'
Set string value... row phrase 'phrase$'
Set string value... row cons 'consonant$'
#Set string value... row vowel2 'vowel2$'
Set string value... row start 'start:6'
Set string value... row end 'end:6'
Set string value... row dur 'duration:6'
#
select TextGrid 'speaker$'
endif
endfor
Remove
endfor
endproc

```

Pronto! A tabela foi criada! O script seguinte adicionará os valores de F0 à Table5600.txt, que já está salva no mesmo diretório onde o script está salvo.

1.3 Adicionar valores de F0

Abra o script AddPitch seguindo o mesmo procedimento para abrir scripts descrito na seção anterior.

Neste script, apenas devem ser alterados o número de linhas da tabela e a porcentagem da área da vogal a ser medida. No atual exemplo, o interesse é medir os 40% centrais das vogais (veja explicação no script).

Script 2 – Adicionar F0

```

# Praat script AddPitch.praat
# Paul Boersma, April 25, 2006

Read Table from table file... table5600.txt
numberOfRows = Get number of rows
assert numberOfRows = 162
previousSpeaker$ = ""
for row to numberOfRows
    speaker$ = Get value... row speaker
    gender$ = Get value... row gender
    start = Get value... row start
    end = Get value... row end
    #
    # Be a bit economical with memory space.
    #
    if speaker$ <> previousSpeaker$
        if previousSpeaker$ <> ""
            select Sound 'previousSpeaker$'
            plus Pitch 'previousSpeaker$'
            Remove
        endif
    endif
endif

```

Alterar o número de linhas.

```

endif
Read from file... 'speaker$'.wav
previousSpeaker$ = speaker$
pitchFloor = if gender$ = "M" then 60 else 120 fi
To Pitch (ac)... 0 pitchFloor 15 no 0.03 0.45 0.01 0.35 0.14 400

endif
duration = end - start
mid = start + duration / 2
startpart = mid - duration / 5
endpart = mid + duration / 5
select Pitch 'speaker$'
medianPitch = Get quantile... startpart endpart 0.5 Hertz
#
# Save results in table5600.
#
select Table table5600
if medianPitch = undefined
    medianPitch = 0
endif
Set string value... row F0 'medianPitch:3'
endfor
Write to table file... table5600.txt
select Sound 'previousSpeaker$'
plus Pitch 'previousSpeaker$'
Remove

```

Área a ser medida: 40% da porção central de cada vogal. A porcentagem pode ser alterada aqui de acordo com o interesse de cada análise.

Adicionados os valores de F0 à tabela, o próximo passo é adicionar os valores dos formantes e largura de banda, como descritos na próxima seção.

1.4 Adicionar formantes e largura de banda

A medição dos formantes também se concentrará nos 40% centrais da vogal usando-se a mesma lógica utilizada no Script 2 (F0).

Script 3 – Adicionar formantes e valores de largura de banda

```

# Praat script AddFormants.praat
# Paul Boersma, April 25, 2006

Read Table from table file... table5600.txt
numberOfRows = Get number of rows
assert numberOfRows = 162
previousSpeaker$ = ""
for row to numberOfRows
    speaker$ = Get value... row speaker
    gender$ = Get value... row gender
    start = Get value... row start
    end = Get value... row end
    #
    # Be a bit economical with memory space.
    #
    if speaker$ <> previousSpeaker$
        if previousSpeaker$ <> ""
            select Sound 'previousSpeaker$'
            Remove
        endif
        Read from file... 'speaker$'.wav
        previousSpeaker$ = speaker$
    else
        select Sound 'speaker$'
    endif
    #
    # Formant analysis.
    #
    formantCeiling = if gender$ = "M" then 5000 else 5500 fi
    duration = end - start
    mid = start + duration / 2
    startpart = mid - duration / 5
    endpart = mid + duration / 5
    Extract part... startpart endpart Rectangular 1.0 no
    Rename... segment
    windowLength = Get total duration
    To Formant (burg)... 0 5 formantCeiling windowLength 50
    for iformant to 3
        f'iformant' = Get value at time... iformant windowLength/2 Hertz Linear
        b'iformant' = Get bandwidth at time... iformant windowLength/2 Hertz Linear
    endfor
    plus Sound segment

```

Alterar o número de linhas.

```

Remove

#
# Save results in table5600.
#
select Table table5600
for iformant to 3
    formant = if f'iformant' = undefined then 0 else f'iformant' fi
    bandwidth = if b'iformant' = undefined then 0 else b'iformant' fi
    Set string value... row F'iformant' 'formant:3'
    Set string value... row B'iformant' 'bandwidth:3'
endfor
endfor
Write to table file... table5600.txt
select Sound 'previousSpeaker$'
Remove

```

Agora a tabela está completa, com todas as colunas preenchidas. Para melhor visualizar os resultados, alguns scripts para plotar vogais serão apresentados na próxima seção.

2. Plotagem de vogais

Antes de plotar as vogais (F1 e F2), é interessante confeccionar uma tabela com as medianas dos valores das vogais. Para tal, o script Table280.praat pode ser rodado. Através dele, apenas os dados das colunas Gender, Dialect, Vowel, Speaker, dur, F1, e F2 serão considerados, e a mediana das três colunas será exibida. Desde que a Table5600 esteja no diretório onde o script foi salvo, não há necessidade de fazer alterações no Script 3, e a nova tabela será salva automaticamente no mesmo diretório.

Script 3 – Otimização da tabela 5600

```

# Praat script, Reduce Table 5600
# Paul Boersma, April 25, 2006

Read Table from table file... table5600.txt
Collapse rows... "gender dialect vowel speaker" "" "" "dur F1 F2" ""
Rename... table280
Formula... dur fixed$ (self, 6)
Formula... F1 fixed$ (self, 3)
Formula... F2 fixed$ (self, 3)
Write to table file... table280.txt

```

Para visualizar a tabela, em Praat objects abra o menu Read, opção Read Table from table file. Selecione esta tabela e clique em Edit.

Para visualizar as vogais em escala linear, rode o seguinte script:

Script 4 – Visualizar as medianas das vogais em um plot

```

# Visualize the vowels on Table 280
# Paul Boersma, April 25, 2006

Read Table from table file... table280.txt
Erase all
Scatter plot... F2 3000 600 F1 1000 200 vowel 12 yes
Write to EPS file... Plot_linear.eps

```

A seguinte figura será plotada e salva no diretório onde o script está salvo:




Figura 4. Plot de vogais (em Hz) em escala linear.

Note que, ao inserir este plot no Word, o símbolo fonético de /æ/ será representado por $\frac{3}{4}$. No entanto, ao transformar o arquivo .doc para PDF, a vogal será plotada corretamente.

Para plotar os valores da Tabela 5600 em escala linear, o Script 5 pode ser utilizado. Diferentemente dos scripts anteriores, este não tem o comando que abre a tabela a partir da qual os dados serão plotados. Então, para abrir a Tabela 5600, em Praat objects escolha no menu Praat a opção, Read Table from table file. Uma vez aberta na janela Praat objects, selecione a Tabela 5600 e rode o Script 5.

Script 5

```
# Ricardo Bion
# January 2006

clearinfo
Erase all

form PARTICIPANT
comment put 0 for all participants
integer plot_participant: 1
integer max_F2: 2800
integer min_F2: 700
integer max_F1: 1000
integer min_F1: 200
endform

select all
tempt = selected("Table")

if plot_participant > 0
Extract rows where column (number)... speaker "equal to" 'plot_participant'
endif

color_of_the_vowel$ = "Black"
line_of_the_sd$ = "Dotted line"

table1 = selected("Table")
Collapse rows... vowel "" "F1 F2" "" ""
nrows = Get number of rows
table2 = selected("Table")

for i from 1 to nrows
select table2
label$ = Get value... i vowel

for formant from 1 to 2
f'formant'_em_Hz = Get value... i F'formant'
endfor

select table1
Extract rows where column (text)... vowel "is equal to" 'label$'

for formant from 1 to 2
sd_F'formant'_em_Hz = Get standard deviation... F'formant'
endfor
```

```

call plot
endfor

select all
minus tempt
Remove

procedure plot

f1 = f1_em_Hz
f2 = f2_em_Hz
stdev_f2 = sd_F2_em_Hz
stdev_f1 = sd_F1_em_Hz

Select outer viewport... 0 10 0 8
Black
Line width... 1
Plain line
Font size... 18
Axes... max_F2 min_F2 max_F1 min_F1

Marks bottom... 5 yes yes no
Marks left... 5 yes yes no
Draw inner box

Text left... yes %F_%1 %(%H%e%r%t%z%)
Text bottom... yes %F_%2 %(%H%e%r%t%z%)

'color_of_the_vowel$'
Text special... 'f2:0' Centre 'f1:0' Half Times 24 0 'label$'
Plain line
Line width... 1

x1 = 'f2'-'stdev_f2'
x2 = 'f2'+stdev_f2'
y1 = 'f1'+stdev_f1'
y2 = 'f1'-'stdev_f1'

'line_of_the_sd$'
Line width... 1
Draw ellipse... 'x1' 'x2' 'y1' 'y2'
endproc

```

Ao rodar este script obtém-se o seguinte plot:




Figura 5. Plot de vogais em escala linear (elipses indicam o desvio padrão).

Note que este plot não foi salvo automaticamente pelo script. Para fazê-lo, na janela Praat picture, escolha no menu File a opção Write to EPS file. Este arquivo pode ser aberto no Word através do comando inserir figura. Sugiro que a figura seja inserida numa caixa de texto do Word.

Para visualizar os mesmos dados em escala logarítmica (log10), rode o Script 6. Como no script anterior, este script não abre a Tabela 5600 e nem salva o plot automaticamente.

Script 6 – Plotar vogais em escala logarítmica

```
# Ricardo Bion
# November, 2006

clearinfo

form PARTICIPANT
comment put 0 for all participants
integer plot_participant: 0
integer max_F2: 3000
integer min_F2: 800
integer max_F1: 1000
integer min_F1: 200
endform

#####
Erase all
Select outer viewport... 0 8 0 6
Black
Line width... 1
Plain line
Font size... 12
Axes... log10(max_F2) log10(min_F2) log10(max_F1) log10(min_F1)

One logarithmic mark bottom... 800 yes yes no
One logarithmic mark bottom... 1000 yes yes no
One logarithmic mark bottom... 1500 yes yes no
One logarithmic mark bottom... 2000 yes yes no
One logarithmic mark bottom... 2700 yes yes no
One logarithmic mark bottom... 3000 yes yes no
One logarithmic mark left... 300 yes yes no
One logarithmic mark left... 400 yes yes no
One logarithmic mark left... 500 yes yes no
One logarithmic mark left... 600 yes yes no
One logarithmic mark left... 800 yes yes no
One logarithmic mark left... 1000 yes yes no

Draw inner box

Text left... yes %F_%1 %(%H%e%r%t%z%)
Text bottom... yes %F_%2 %(%H%e%r%t%z%)

#####

select all

tempt = selected("Table")

Copy... tempt
Formula... F1 log10(self)
Formula... F2 log10(self)

if plot_participant > 0
Extract rows where column (number)... speaker "equal to" 'plot_participant'
endif

color_of_the_vowel$ = "Black"
line_of_the_sd$ = "Plain line"

table1 = selected("Table")
Collapse rows... vowel "" "F1 F2" "" ""
nrows = Get number of rows
table2 = selected("Table")

for i from 1 to nrows
select table2
label$ = Get value... i vowel

for formant from 1 to 2
f'formant'_em_Hz = Get value... i F'formant'
endfor

select table1
```

Nestes campos define-se os valores mínimos e máximos de F1 e F2 que serão utilizados no frame do plot. Eles são ajustados de acordo com as vogais ou sexo dos participantes.

Nesta parte definem-se quantas e quais marcações dos valores em Hz aparecerão no frame do plot.

Aqui define-se a cor da vogal e o estilo de linha que indicará o desvio padrão. As opções de cores e estilos de linha estão listadas em Praat picture, menu Pen.

```

Extract rows where column (text)... vowel "is equal to" 'label$'

for formant from 1 to 2
sd_F'formant'_em_Hz = Get standard deviation... F'formant'
endfor

call plot

endfor

select all
minus tempt
Remove

procedure plot
f1 = f1_em_Hz
f2 = f2_em_Hz
stdev_f2 = sd_F2_em_Hz
stdev_f1 = sd_F1_em_Hz

'color_of_the_vowel$'
Text special... 'f2' Centre 'f1' Half Times 24 0 'label$'
Plain line
Line width... 1

x1 = 'f2'-'stdev_f2'
x2 = 'f2'+ 'stdev_f2'
y1 = 'f1'+ 'stdev_f1'
y2 = 'f1'-'stdev_f1'

'line_of_the_sd$'
Line width... 1
Draw ellipse... 'x1' 'x2' 'y1' 'y2'
endproc

```

Este script plotará o seguinte:




Figura 6. Plot de vogais em escala logarítmica.

3. Comentários finais

Para ter acesso ao arquivo de som que serve de exemplo neste tutorial, acesse www.nupffale.ufsc.br/rauber. Você encontrará o link *Example - wav* no canto inferior esquerdo da página. Os demais arquivos que contêm os scripts, tabelas e plots poderão ser solicitados através do e-mail asrauber@gmail.com, o que facilita o acesso aos mesmos para que sejam alterados de acordo com os interesses de cada pesquisador.

Bibliografia

Boersma, P., & Weenink, D. (2006). *PRAAT: doing phonetics by computer (Version 4.4.23)* (Computer program), retrieved 12 June 2006, from <http://www.praat.org>.